### Coding Guidelines for Prolog

#### MICHAEL A. COVINGTON

Institute for Artificial Intelligence, The University of Georgia, Athens, Georgia, U.S.A. (e-mail: mcQuga.edu)

#### ROBERTO BAGNARA\*

Department of Mathematics, University of Parma, Italy (e-mail: bagnara@cs.unipr.it)

#### RICHARD A. O'KEEFE

Department of Computer Science, University of Otago, Dunedin, New Zealand (e-mail: ok@cs.otago.ac.nz)

#### JAN WIELEMAKER

Department of Computer Science, VU University Amsterdam, The Netherlands (e-mail: j.wielemaker@uva.nl)

#### SIMON PRICE

Department of Computer Science, University of Bristol, United Kingdom (e-mail: simon.price@bristol.ac.uk)

submitted 15 November 2009; revised (not yet); accepted (not yet)

#### Abstract

Coding standards and good practices are important for all software projects, whatever programming languages they employ. Arguably, Prolog programming can benefit from a disciplined approach more than programming in other languages. Despite this, no widely accepted standards and practices seem to have emerged up to now. The present paper aims at being a first step towards filling this gap.

Keywords: Prolog, style, coding standards, debugging, efficiency

#### 1 Introduction

The purpose of programming languages is to make programs readable by human beings. Coding standards enhance this function, especially when multiple programmers and a need for maintainability are present, but also even in the small projects of a single programmer (one must, after all, read and debug one's own work). Coding standards for Prolog are particularly needed for several reasons:

<sup>\*</sup> The work of R. Bagnara has been partly supported by PRIN project "AIDA2007 — Abstract Interpretation Design and Applications."

Availability. As far as we know, a coherent and reasonably complete set of coding guidelines for Prolog has never been published. Moreover, looking at the corpus of the published Prolog programs, no de facto standard seems to have emerged. The most important reason behind this apparent omission is that the small Prolog community, due to the lack of a comprehensive language standard, is further fragmented into sub-communities centered around individual Prolog systems, none of which has a dominant position (contrast this with the situation of Java and Sun's coding conventions for that language (Sun Microsystems, Inc. 1999) or the precedents set for C by the UNIX source code).

Language. Language features that contribute to the power of the language make it quite easy —especially for the non-expert— to get things wrong, sometimes in ways that are difficult to diagnose. For example, the lack of prescriptive typing contributes to the suitability of Prolog for quick prototyping in some application domains. At the same time, the fact that no redundant type information is available to the development tools makes discipline necessary. Besides types, Prolog developers and maintainers are confronted with modes: the same arguments of procedures can be inputs, outputs, or both. This gives Prolog conciseness and elegance through reversible predicates but makes it necessary to keep track of which modes are supposed to be supported (within reasonable computational complexity limits) by which predicates.

Compiler technology. Most Prolog compilers give no style warnings except for singleton variables. Even though more advanced tools and development environments exist, a disciplined approach is still the best device available to those programming in standard Prolog. Indeed, one can argue that adherence to coding standards enhances the power of any programming language.

It is important to stress that we are not announcing the coding standard for Prolog. This is a paper with five authors and, on some points, more than five opinions. Rather, we address numerous issues that the maker of a full-fledged coding standard will have to confront. In some cases, there are good reasons to prefer one alternative to another. Other decisions are arbitrary, like driving on the left or on the right; a community of programmers can choose any of several ways of doing something as long as they are consistent.

The amount of standardization needed depends on the scale and duration of the programming project. Just like a novel or a scientific paper, the quality of a computer program can only be judged from the degree to which it reaches its objectives, i.e., "it works": for example, we might say that a novel works if it becomes a best-seller, the manual of a device works if it enables the average user to effectively operate the device, a scientific paper works if it is published in a prestigious journal. Computer programs written for different purposes may or may not "work" in different ways. Only for very simple programs written to solve very simple tasks we can say that the program works if it functions correctly. For even moderately complex programs a more sensible definition could be: the program works if its developers and maintainers are able to approach the correct expected behavior of the program over its intended lifespan.

In some cases correctness and maintainability over a long period of time are not important. For instance, in rapid prototyping of applications (something for which Prolog has its advantages) we might say that the program (prototype) works if (despite its errors and limitations) it proves that the approach is feasible. Still, the prototype will be incomplete until it is finished, and its chances of being finished one day (as opposed to collapsing and being abandoned before being of any use) depend on qualities such as readability, extensibility, and whatever else helps the development team.

In a sense, the text of a program is not very different from an argumentative essay. Like an argumentative essay, a nontrivial program is addressed to an *audience*, which needs to be accurately identified for the argumentation part of the program to be effective. As for any argumentation, different audiences will require and be prepared to share a different set of *premises* (hypotheses and preconditions explicitly mentioned in the program text) and *assumptions* (hypotheses and knowledge that are left implicit but that still are essential for the comprehension of the program). Not to mention that, in programming, the intended audience is also heavily influenced by the tools it is used to.

To summarize, the existence of very different purposes and wildly different audiences are such that a full-fledged coding standard can only be decided upon on a per-project basis. Moreover, different, equally reasonable and effective coding standards can, on specific points, recommend plainly opposite things: a coherent whole matters more than the individual bits.

In this paper, which began as an evolution of (Covington 2002), we introduce a set of coding guidelines that can serve as a starting point for the development of effective coding standards. We highlight the aspects of Prolog program development that deserve particular attention and can benefit from a disciplined approach; we illustrate the rationale that is behind each of the proposed guidelines; when alternative guidelines can achieve the desired effect, their relative merits are discussed.

The paper is organized as follows: in Sections 2, 3 and 4 we discuss guidelines concerning the naming of program entities, code layout and documentation, respectively; Section 5 concerns the effective use of language features; Section 6 deals with the development, debugging and testing of program units and their interfaces; Section 7 concludes.

#### 2 Layout

Do not let your programs be hampered by inconsistent layout. Poor layout is painful to work with, is distracting, and can lead to otherwise perfectly avoidable mistakes. Moreover, most text editors, if properly operated, can provide significant assistance in ensuring layout consistency.

#### 2.1 Indent with spaces instead of tabs.

You cannot normally predict how tabs will be set in the editors and browsers others will use on your code. Moreover, mixing tabs and spaces makes searches and

substitutions more problematic. All in all, it is better not to use tabs altogether: almost any editor can be set to substitute spaces for tabs automatically when saving your file.

#### 2.2 Indent consistently.

Code is more readable if it is indented consistently. Even though the choice of the "right" indent size often leads to friction within the development team, a decision has to be made. Most editors default to an indent size of 8 and while such a large indent does make shallowly nested indentation obvious, it uses up excessive horizontal screen real estate and makes even moderately nested indentation hard to read. A small indent size of 2 spaces is popular with good Prolog programmers but can make it hard for less keen eyed readers or for those using variable pitched fonts to interpret the indentation depth. The compromise value of 4 has proven popular and practical in both Prolog and other well known languages.

If in doubt: Use an indent size of 4 spaces.

#### 2.3 Limit the length of source code lines.

Lines longer than 80 characters are almost always difficult to read. While 80-column-wide screens are no longer in widespread use, 80 columns is still the default width for many text editors. Some editors may also display the last column in an inconvenient way, so limiting the line length to 79 or 78 characters is a good idea.

High-quality printed listings can usually accommodate no more than 65 or 70 characters per line, depending on the type size and paper size. For maximum readability you may want to lower this limit down to 55 characters (Covington 1994).

#### 2.4 Limit the length (number of lines) of clauses.

The ideal situation would of course be for each clause to fit onto an ordinary computer screen. So 24 lines is the safest choice but lengths up to 48 can fit on modern computer displays. You can consider this a limit not to be crossed, the exceptions being for predicates doing long but conceptually simple sequences of I/O or graphic operations.

More generally, always consider simplifying long clauses, whether or not they fit onto your screen. Since in Prolog the only way to have a loop is to introduce another predicate, and case analyses often (though not always) deserve their own predicates too, it is no hardship to follow this guideline.

#### 2.5 Be consistent in the use of space around commas.

The simplest approach is to follow each comma with a space, just as in English. This improves readability by increasing the visual separation among different arguments of a term. Moreover, as this is consistent with the typesetting conventions used for

many other languages, doing so often avoids a disturbing mismatch between the program text, its comments, and the string literals used in the program.

The simplicity of the above guideline is valuable, but it misses the opportunity to reduce the decoding effort for readers. An alternative is thus to use space in a way that helps in recognizing the various uses of commas.<sup>2</sup> One possibility is to distinguish between:

- comma meaning *and-then*: follow it by one or more spaces, an optional comment and a newline (see also Guideline 2.7);
- comma separating arguments of a goal: follow it by one or more spaces;
- comma separating elements of a data structure: do not follow it by spaces.

Of course, in order to break long lines, extra newlines should be allowed in the second and third case. In addition, it is tempting to use the ability of putting one or more spaces to line stuff up, as in:

Similarly, it may seem natural (to one person, on a particular day, and in a particular context) to relax the rules in order to help distinguishing "levels", as in [[1,2,3], [4,5], [6,7,8,9]]. Note also that [alpha, beta, gamma] looks nice because the terms are longer than in [1,2,3], but if you allow that to make a difference, you can no longer rely on spaces to tell you anything you want to know.

This is why layout remains an art rather than a science. There will always be a friction between the rules —which tend to be either too complex or too rigid but allow for uniformity— and taste —which can adapt to particular situations in often nicer ways but is personal, variable and sometimes inconsistent.

For small individual projects, following some general guidelines with flexibility and applying taste on a case-by-case basis can give good results. However, for projects that are bigger and/or have several developers it is best to take style decisions on the basis of explicit and precise principles rather than on the basis of what feels good at the time: sticking to precise rules may occasionally lead to some program fragments that do not look very good, but the overall readability of the project as a whole will benefit.

### 2.6 Begin each clause on a new line and indent all but the first line of each clause.

For example, write:

<sup>&</sup>lt;sup>1</sup> Both natural and formal languages, such as Ada 95 (Ausnit-Hood et al. 1997) and Java (Sun Microsystems, Inc. 1999).

<sup>&</sup>lt;sup>2</sup> It was pointed out long ago by critics of Prolog that different commas mean different things. Of course, Prolog is by no means alone in this: take C as an example.

```
same_length([], []).
same_length([_|L1], [_|L2]) :-
same_length(L1, L2).
```

#### 2.7 Put each subgoal on a separate line.

Putting each subgoal on a new line greatly enhances readability. So, no matter how short some subgoals may be, it is better to use one line for each of them. For example:

```
ord_union_all(N, Sets0, Union, Sets) :-
   A is N/2,
   Z is N-A,
   ord_union_all(A, Sets0, X, Sets1),
   ord_union_all(Z, Sets1, Y, Sets),
   ord_union(X, Y, Union).
```

The only exception might be for short sequences of closely related subgoals, such as those involving write and nl. For instance, one could have

```
write('CPU time = '), write(T), write(' msec'), nl.
```

Note though that the format/[1,2,3] built-in predicates (not yet in the Prolog standard, but provided by quite a few Prolog systems with reasonably compatible implementations) provides a better way to output several items in a single conceptual step.

#### 2.8 Use vertical space consistently to improve readability.

It is natural for vertical distance to reflect the logical distance. For example: begin next clause of the same predicate on next line; skip a line before the first clause of the next related predicate; skip two lines before the first clause of the next unrelated predicate. Notice that the unindented clause head is enough to separate clauses within the definition of a single predicate: no lines need to be skipped here.

#### 2.9 Comment source files, not just the predicates within them.

In addition to the comments on particular clauses or predicates, a Prolog source file should begin with some sort of standard header with meta-data about the file: name, version, author(s), revision history, copyright, license and other information. Next there should be the module declaration (in case a module system is used) and maybe other declarations. Immediately after the declaration a detailed comment should explain what the file is all about, what the central ideas are, possibly with illustrations of the data structures used.

It is best if this long explanatory comment is readable as plain text with a minimum of distracting punctuation: it is thus better to use plain /\* and \*/ delimiters

with no extra stars.<sup>3</sup> (The % comment delimiter is reserved for comments pertaining to individual predicates, as described below.) If a file divides naturally into several sections, each section can have its own such comment.

#### 2.10 Use layout to make comments more readable.

While text editors can provide significant help in properly formatting code, for comments the developer is usually on his or her own. Nonetheless, a little extra care can do much to increase the readability and usefulness of comments. For example, instead of writing:

```
% This predicate classifies C as whitespace (ASCII < 33), alphabetic
% (a-z, A-Z), numeric (0-9), or symbolic (all other characters).
write something like this:

% This predicate classifies C as:
% - whitespace (ASCII < 33);
% - alphabetic (a-z, A-Z);
% - numeric (0-9); or
% - symbolic (all other characters).</pre>
```

Indented lists like this are much easier to read than lists disguised as paragraphs. It is also much less work to add or remove an item or add comments about an item.

## 2.11 Avoid comments to the right of the code, unless they are inseparable from the lines on which they appear.

Comments on the right are problematic, since they so easily get out of step with the code when small changes are made to the program. Moreover, as soon as there is more than one end-of-line comment in the same predicate, it looks bad if they do not begin in the same column, and maintaining this alignment when editing the program becomes an extra chore.

If you cannot dispense with comments to the right of the code, at least make them very short: just a few crucial hints and, if needed, a pointer to the predicate documentation. For example:

```
%% pred(...)
% ...
% [Note 1] Long comment about how pred/n depends on subgoal/k...
% ...
pred(...) :-
    ...
    subgoal(...),  % See [Note 1] above.
```

 $<sup>^{3}</sup>$  But possibly with dashes at the top and bottom to increase visual separation.

#### 2.12 Consider using comments consistently as reminders.

With some discipline you can effectively mark the points that require your attention at a later time. One possibility is to use the % comment delimiter to create "tags" that can then easily and reliably retrieved with string search. For example:

```
%TBD: <short hint on what remains to be done>
%FIXME: <what/why this does not work in all cases>
%HACK: <why this is not completely satisfactory>
```

Here TBD abbreviates to be done. These abbreviations correspond to the //TODO: comments familiar to C# and Java programmers in various development environments.

In more extreme cases, you can guarantee that an unfinished section of your program is not used by inserting Prolog code that will not work. In an unfinished predicate, insert a call to something like tbd('description of what is to be done') and do not define the predicate tbd. Then any attempt to call the predicate will throw an error.

#### 2.13 Make clauses understandable in isolation.

As far as you can easily do so, enable the reader to understand each clause without having read everything above it. For example, if a test is unnecessary because a cut guarantees its truth, say so in a comment at the appropriate place. For example, write:

```
remove_duplicates([First|Rest], Result) :-
    member(First, Rest),
   !,
   remove_duplicates(Rest, Result).
remove_duplicates([First|Rest], [First|New_Rest]) :-
   % First does not occur in Rest.
   remove_duplicates(Rest, New_Rest).
```

The comment keeps the clause from being misunderstood if the reader has not read the preceding clause.

### 2.14 Indent an additional level between repeat and the corresponding cut.

This makes a repeat structure look more like a loop. Here is an example:

```
process_queries :-
    repeat,
        read_query(Q),
        handle(Q),
        Q = [quit],
!,
write('All done.'), nl.
```

#### 2.15 Decide how you want to break long clause heads and subgoals.

Even if you try to limit the number of arguments of your predicates (see Guideline 5.3), and, or perhaps because, you choose your names wisely (see Guidelines 3.1–3.12), you will occasionally end up with a clause head or a subgoal that is too long to fit on a single line. In this case, two alternatives make sense:

The second alternative may be perceived as less pretty, but it keeps predicate name length from affecting indentation. By contrast, the first alternative forces you to change the indentation of long\_arg4 every time the predicate is renamed (which happens fairly often during development). However, the lengths of the arguments still affect layout. Alternatives include placing each argument on its own line or placing only semantically related groups of arguments on the same line.

#### 2.16 Decide how to format disjunctions and if-then-elses.

Prolog programmers are deeply divided on how to format the semicolon (which means "or") and the if-then-else structure (Goal1 -> Goal2; Goal3). Above all, note that:

- Heavy use of semicolons and if-then-elses may indicate that the program is not broken up into clauses correctly. The main ways to indicate disjunction and conditionality are to provide multiple clauses and control the success conditions of each clause. The semicolon and if-then-else provide alternatives for breaking out of the usual Prolog program structure.
- A semicolon at the end of a line will be ignored and will surely lead to error.
- Programmers do not remember the relative precedence of commas and semicolons; a,b;c means (a,b);c but this fact is not obvious. Thus, layout and parentheses to indicate precedence are recommended.

All but the simplest disjunctions should be displayed prominently using layouts such as

```
disjunct_1
;
disjunct_2
```

```
;
    disjunct_3
)
or the more compact
(    disjunct_1
;    disjunct_2
;    disjunct_3
)
```

In both styles, parentheses are always present, the closing parenthesis is exactly below the opening one, and the semicolons stick out in a way that makes misunder-standing much more difficult. If all of this seems too prolix, recall the first bulleted item above.

The second style, besides compactness, has the advantage that it naturally generalizes to if-then-elses as follows:

```
( test_1 ->
    test_1_is_true
; test_2 ->
    test_2_is_true_and_test_1_is_false
; both_are_false
)
```

A variation that lets the '->' operators be more visible at the expense of more vertical space usage is:

```
(
    test_1
->
    test_1_is_true
;
    test_2
->
    test_2_is_true_and_test_1_is_false
;
    both_are_false
)
```

This can also be written as

```
(test_1 ->
     test_1_is_true
;
     (test_2 ->
           test_2_is_true_and_test_1_is_false
;
     both_are_false
```

```
)
```

which is more expensive both in terms of horizontal and vertical space, but some consider it more readable. A third alternative is to put -> and ; at the beginnings of lines:

```
(test_1
-> test_1_is_true
; (test_2
    -> test_2_is_true_and_test_1_is_false
    ; both_are_false
    )
)
```

A further variation is to put each -> and; on a line by itself. Whichever style you choose, you should work out how it will handle nesting.

#### 2.17 Consider using automated tools to improve readability.

A pretty-printer is a program that prints your programs in a neat, readable form. Pretty-printers are sometimes called *code beautifiers*.

To be more precise, we should recall that, historically, a "pretty-printer" was a program that altered the white space in the source code (adding and removing spaces and line breaks) in order to achieve consistent indentation. For C, cb and GNU's indent are programs of this type. Some Lisp and Scheme systems have a pp function for printing functions prettily. Pretty-printing of this type can greatly improve code readability.

Automatic (re-)indentation is built into many editors, as is the ability to alter the type style (plain, italic, bold, underlined, etc.) and/or the colour of tokens in order to indicate their role in the syntax (such as which characters are part of comments, which are part of strings, and which are other program text). An example is Emacs' electric font lock. For a token styler that handles several programming languages, including Prolog, see m2h by Richard O'Keefe (http://www.cs.otago.ac.nz/staffpriv/ok/software.htm).

Nowadays, the term "pretty-printing" refers mainly to hardcopy and to listings that are embedded in documentation, textbooks, and the like. For example, consider the code in Figure 1. The LATEX listings package, which is highly customizable and included in all LATEX distributions along with documentation, pretty-prints Prolog code that is surrounded with the lines:

```
\lstset{language=Prolog, frame=lines}
\begin{lstlisting}[caption={Pretty-printing example},label=pp-ex]
...
\end{lstlisting}
```

The result is shown in Figure 2. Note how the characters of proportional-pitch type are spread apart to help maintain horizontal alignment.

```
%% sum_list(+Numbers_List, ?Result)
   Unifies Result with the sum the numbers in Numbers_List;
   calls error/1 if Numbers_List is not a list of numbers.
sum_list(Numbers_List, Result) :-
    sum_list(Numbers_List, 0, Result).
% sum_list(+Numbers_List, +Accumulator, ?Result)
sum_list([], A, A).
                          % At end: unify with accumulator.
sum_list([H|T], A, R) :- % Accumulate first and recur.
 number(H),
 !,
 B is A+H,
 sum_list(Rest, B, R).
sum_list(_, _A, _R) :-
                          % Catch ill-formed arguments.
  error('invalid first arg to sum_list/2').
```

Fig. 1. Prolog code to be pretty-printed

```
%% sum\_list(+Numbers\_List, ?Result)
    Unifies Result with the sum the numbers in Numbers_List;
%
    calls error/1 if Numbers_List is not a list of numbers.
sum_list(Numbers_List, Result) :-
    sum_list(Numbers_List, 0, Result).
  sum\_list(+Numbers\_List, +Accumulator, ?Result)
sum_list([], A, A).
                           % At end: unify with accumulator.
sum\_list([H|T], A, R) :- \% Accumulate first and recur.
 number (H),
  !,
 B is A+H,
  sum_list(Rest, B, R).
sum_list(\_, \_A, \_R) :=
                           \% Catch ill-formed arguments.
  error ('invalid_first_arg_to_sum_list/2').
```

Fig. 2. Prolog listing formatted by LATEX listings package

An alternative approach to presenting Prolog programs elegantly in IATEX is pltex by Michael Covington (http://www.ai.uga.edu/mc/pltex.zip). This experimental program renders the same Prolog example as shown in Figure 3. Note the use of font styles to indicate syntax, and the replacement of :- by ←. The approach is inspired by "Algol style" in early computer programming literature (Covington 1994).

Also, the aforementioned m2h program can produce LaTeX output, in addition to HTML and other output formats. The documentation of the LaTeX listings

```
%% sum_list(+Numbers_List, ?Result)
% Unifies Result with the sum the numbers in Numbers_List;
   calls error/1 if Numbers_List is not a list of numbers.
sum_list(Numbers_List, Result) \leftarrow
     sum\_list(Numbers\_List, 0, Result).
% sum_list(+Numbers_List, +Accumulator, ?Result)
                                   % At end: unify with accumulator.
\mathbf{sum\_list}([], A, A).
\mathbf{sum\_list}([H|T], A, R) \leftarrow
                                   % Accumulate first and recur.
  number(H),
  B is A+H,
  \mathbf{sum\_list}(Rest, B, R).
                                   % Catch ill-formed arguments.
\mathbf{sum\_list}(\_,\_A,\_R) \leftarrow
  error('invalid_first_arg_to_sum_list/2').
```

Fig. 3. Prolog listing formatted by Covington's pltex

package gives references to other pretty-printers, several of which are applicable to Prolog.

Programmers are lulled into complacency by conventions. By every once in a while, by subtly violating convention, you force [the maintenance programmer] to read every line of your code with a magnifying glass.  $\qquad \qquad - \text{ROEDY GREEN (Green 2008)}$ 

#### 3 Naming Conventions

Choosing the right names is among the most crucial activities undertaken by any software developer, especially when the programming language is Prolog (which is untyped, with parameters that can act as input, output or input/output and so forth). Detailed advice on naming can be found in (Ledgard and Tauer 1987) and (Kernighan and Pike 1999, p. 104 ff.).

#### 3.1 Choose a writing style for multiple-word identifiers.

The basic choice is between underscore-style (like\_this) and internal capitalization ("intercaps," LikeThis). A complicating factor is that Prolog requires variables to begin with an uppercase letter and atoms to begin with a lowercase letter. This tempts some programmers to use "camel case" (likeThis), i.e., intercaps with the first letter lowercased, which other programmers find particularly unaesthetic.

Recall that intercaps became popular in programming languages that did not allow underscores within names, such as Pascal and Smalltalk. Arguably, in using intercaps, one seems to be reverting to the early Middle Ages, when handwritten words were not separated (Thompson 1893, page 65), and thus flouting an important readability tool that is a thousand years old.

Since Prolog allows underscores, and underscores resemble spaces, there is much to be said for using underscores everywhere a word separator is needed, both in atoms and in variable names. The question then remains how to capitalize variables and atoms. The possibilities are to stick with one case, as in Result\_So\_Far and is\_boolean\_function, or to choose the most appropriate case for each individual word in a compound identifier, as in Result\_so\_far and is\_Boolean\_function. The former convention is certainly more appropriate for variable names, with the only possible exception constituted by short suffixes a program might consistently use to make the information flow explicit, as in

```
simplify_expressions([E_in|Es_in], [E_out|Es_out]) :-
    simplify_expression(E_in, E_out),
    simplify_expressions(Es_in, Es_out).
```

An alternative practice, recommended by some, is to use intercaps for variable names and underscores within atoms. In any case, consistency must be ensured: knowing the words used in an an identifier and knowing whether it is an atom or a variable should allow any team member to know how to spell it immediately.

If in doubt: Use underscores to separate words in compound identifiers, i.e., write is\_well\_formed, not isWellFormed. Prefer Result\_So\_Far to Result\_so\_far.

#### 3.2 Make all names pronounceable.

A predicate named stlacie is going to confuse anyone else reading your program—including yourself at a later date— even though it may seem, at the time of writing, to be a perfectly obvious way to abbreviate "sort the list and count its elements."

If you don't enjoy typing long names, type short ones. Call your predicate sac or even sc while typing in your program, then do a global search-and-replace to change it to sort\_and\_count.

#### 3.3 Avoid using different names that are likely to be pronounced alike.

If you use foo, do not also use fou or fu. People remember pronunciations, not spellings. Accordingly, it must be absolutely obvious how to spell a name when all you remember is its pronunciation.

For the same reason, do not mix up to, two, and too. At one time it was fashionable to abbreviate "to" as "2," thereby saving one character. That is how computer programs got names such as afm2tfm (for "AFM-to-TFM," a  $T_EX$  utility) and DOS's exe2bin. However, this practice creates too much confusion. Remembering how to spell words correctly is hard enough: do not ask readers of your code to remember your creative misspellings, too. With the exception of the widely used i18n (for "internationalization," a usage coined at DEC many years ago) and L10n (for "localization"), spellings like 18tr (or 18r?) and v1r3d do not facilitate communication; they just make the reader suspect that you are still in high school.

#### 3.4 Within names, do not express numbers as words.

If you have three predicates for which you have no better names, call them pred1, pred2, and pred3 — not pred\_one, pred\_two, and pred\_three. This is yet another stratagem to make spellings more predictable from pronunciations.

Furthermore, exported predicates should not have numeric suffixes unless the number is some sort of code number. For example,

```
unicode_4_0_0(?Code, ?Class) might be reasonable.
```

#### 3.5 Choose sensible names for auxiliary predicates.

If part of the algorithm for your predicate, which you have named foo, needs to be placed in another predicate definition, do not immediately call that predicate foo\_aux; usually there are better alternatives. For example:

- If the auxiliary predicate and the main predicate have different numbers of arguments, their names can be the same.
- If an auxiliary predicate is there to do a case analysis, you can use \_case as a better suffix than \_aux.
- If an auxiliary predicate is a loop, \_loop may be a good suffix.
- If an auxiliary predicate relies on a higher predicate to take care of resource allocation and disposal, \_unguarded might be a good suffix, as in:

```
foo(...) :-
    acquire_resources(...),
    call_cleanup(foo_unguarded(...), release_resources(...)).
foo_unguarded(...) :-
```

When all else seems inappropriate \_aux can be used. Other arbitrary alternatives are \_1, \_2, \_3 and so on or \_x, \_xx and so forth, but these give up any opportunity to indicate the function of the auxiliary predicate and so should preferably be accompanied by explanatory comments.

## 3.6 If a predicate represents a property or relation, its name should be a noun, noun phrase, adjective, prepositional phrase, or indicative verb phrase.

Examples include:

- sorted\_list, well\_formed\_tree, parent (nouns or noun phrases);
- well\_formed, ascending (adjectives);
- in\_tree, between\_limits (prepositional phrases);
- contains\_duplicates, has\_sublists (indicative verb phrases).

# 3.7 If a predicate is understood procedurally —that is, its job is to do something, rather than to verify a property— its name should be an imperative verb phrase.

Examples include remove\_duplicates (i.e., the second-person imperative form is used, not removes\_duplicates) and print\_contents (not prints\_contents).

#### 3.8 Choose predicate names to help show the argument order.

For example, mother\_of(A,B) is ambiguous; does it mean "A is the mother of B" or "the mother of A is B"? Naming it mother\_child or mother\_of\_child would eliminate the ambiguity.

In general, the predicate name is the only place in Prolog where you can conveniently give the user hints about the argument order. For example, a specification like tree\_size(+T, ?S) makes it clear that the tree argument precedes the size argument. This means that sometimes the most salient type name needs to come last. For example, to convert lists to trees, we might use list\_to\_tree(+L, ?T). Module prefixes, on the other hand, can only be prefixes. If you see a goal of the form tree:from\_list(X, Y), can you tell which of the arguments is the tree and which the list?

### 3.9 Use descriptive names for variables wherever possible, and make them accurate.

For example, do not call a variable Tree if it is not a tree. You would be surprised at how often such things are done by programmers who have changed their minds midway through constructing a clause.

More generally, there are two things that can be encoded in the variable name: its type and its role in the predicate. You can name the variable after only one of these properties if the other is obvious from the context. For example, if you have a predicate graph\_nodes(+Graph, -Nodes), it is already obvious that the role of the second argument is to represent graph nodes, so you may name it List, if it is a list, or Assoc, if it is an assoc; for maximum clarity you may even name it Nodes\_List or Nodes\_Assoc.

#### 3.10 Use single letter names only in presence of suitable conventions.

Prolog has no global variables; most variables exist only in a very limited context. For conciseness, some programmers adopt program-wide conventions whereby, for instance, N is always the total number of elements to process and I is always the number of elements processed so far. If you have a convention that L and U are always the lower and upper bounds of some range, fine. If you have a convention that C, C0, C1, C2 are always character codes, S, S0, S1, S2, are sequences, and so on, fine; make sure that such conventions are prominently stated in the source file documentation or are absolutely obvious from context. Whenever you have the

slightest doubt that the code may be tricky to read, forget about single letter names and give your reader some help.

#### 3.11 Consistently name threaded state variables.

There is a special case for variables that receive an initial state, go through some translations, and come to a final output. It is important to name those consistently. For example, one can use names of the forms StateO (for the initial state), State1, State2 and so on (for the intermediate states), and State (for the final state):

```
foo(..., State0, State) :-
  foo_step(State0, State1),
  foo(..., State1, State).
```

An alternative choice is to use the names State\_in (for the initial state), State\_tmp, State\_tmp1, State\_tmp2 and so forth (for the intermediate states), and State\_out (for the final state):

```
foo(..., State_in, State_out) :-
  foo_step(State_in, State_tmp),
  foo(..., State_tmp, State_out).
```

### 3.12 Use a singular noun for the first element of a list and its plural for the remaining elements.

For example, match a list of trees to [Tree|Trees]. If you use a single letter for a list element, use that letter with an 's' after it for the remaining elements. For example, match a list of trees to [T|Ts]. Only do this if something nearby in the context, such as an adequately detailed comment, makes it obvious what the single letter stands for. A compromise, which may be more readable in some situations, is [T|Trees].

If there are two entities for which you would like to use the same single letter, give them both full names, as in

```
term_translations([], []).
term_translations([Term|Terms], [Translation|Translations]) :-
    term_translation(Term, Translation),
    term_translations(Terms, Translations).
```

Here, using T would be confusing.

### 3.13 Decide whether predicate names should carry the types on which they operate

The reason a decision is necessary is that, when you also have module names, you can end up with redundancies such as quaternion:quaternion\_magnitude(Q, A). If you are using a module system, and you are always including the module name, then you need not put the same information in the predicate names.

On the other hand, a number of of widely accepted libraries use predicate names that carry the types on which they operate. This is the case, e.g., for the ordered set library (using the ord\_ prefix) and the association lists library (using the \_assoc suffix). This convention predates the module system and is similar to a long-standing convention in C, also predating other methods of managing the namespace.

Even when modules are used, some people like to import predicates from other modules and use them without module prefixes, just like you use predicates provided by the system. This is the historically preferred approach, because people expected to convert Edinburgh Prolog code that used plain files to code that used modules just by adding :- use\_module directives. In this style, type names are still useful even if they are identical to the module name.

#### 3.14 Do not blindly import styles from other languages.

For instance, do not blindly follow the Java tradition by calling everything in sight <code>is\_xxx</code> or <code>get\_yyy</code>. It is fine to name <code>is\_tree/1</code> a *checker*, that is, a predicate that will only succeed, without leaving choice-points, on well-formed, structurally complete trees. This will help the user to distinguish it from the *generator* — whether its existence in the program is real or only conceivable—called <code>tree/1</code>.

Similarly, stay away from the massive overuse of "get" that stems from the Java world. It is fine to use "get" to signal that information is being obtained in an extra-logical way, such as when I/O or the foreign language interface are involved. But do not use "get" for fetching a property via an ordinary logical relation.

Much of the skill in writing unmaintainable code is the art of naming variables and methods.

They don't matter at all to the compiler.

That gives you huge latitude to use them to befuddle the maintenance programmer.

— ROEDY GREEN (Green 2008)

#### 4 Documentation

All the good reasons why documentation is crucial to the success of any software project, independently from the languages used to code it, are valid for Prolog code. In addition, features like multiple modes, dynamic code, meta-programming facilities, and the lack of prescriptive typing make documentation even more important than for other programming languages.

On the other hand, writing good documentation is expensive and this cost may not be fully justified for those projects where Prolog is used for rapid prototyping. Each project should choose a consistent documentation standard based on the expected benefit/cost ratio. In this choice an important role can be played by the available tools: if the code is written with a system that provides good support for writing, checking, formatting and retrieving the documentation, it is quite likely

Fig. 4. Example documentation template

that choosing the documentation standard(s) dictated by that system will be the most economical solution.

In this section, we will review the important points to consider when designing a code documentation standard.

### 4.1 Begin every predicate (except perhaps auxiliary predicates) with an introductory comment in a well-defined format.

Predicates that can be called from elsewhere in the program —from code written by others, or by yourself on a different occasion— must be documented by an introductory comment. Here is an example:

```
%% remove_duplicates(+List, -Processed_List) is det
%
% Removes the duplicates in List, giving Processed_List.
% Elements are considered to match if they can
% be unified with each other; thus, a partly uninstantiated
% element may become further instantiated during testing.
% If several elements match, the last of them is preserved.
```

This comment follows the template in Figure 4. Note the overall layout and the use use of %% to mark the first line. Arguments are represented by variable names preceded by mode specifiers from one of the systems in Table 1, and optionally followed by types. At the end of the first line is a determinism specifier from Table 2.

The first of the three systems in Table 1 is preferred because it is the most explicit about flow of information. For example, note that it would be wrong to write

```
compare(?R:order, +T1:term, +T2:term)
```

to document the compare/3 primitive, because T1 and T2 are allowed to be variables, and

```
compare(?R:order, ?T1:term, ?T2:term)
```

would be misleading because we expect that? arguments will be unified with something. The correct annotation in this scheme is thus

```
compare(?R:order, =T1:term, =T2:term)
```

Table 1. Three systems of argument-mode specifiers

#### Recommended system:

- \* ground on entry, i.e, must already be instantiated to a term that contains no free variables when the predicate is called.
- + nonvar on entry, i.e., must already be instantiated to a term that is not a free variable (although the term may contain a free variable) when the predicate is called.
- var on entry, i.e., must be uninstantiated when the predicate is called.
- > Thought of as *output* but might be nonvar.
- = Thought of as *input* but is not modified.
- Not specified, i.e., may or may not be instantiated.

#### System used in PlDoc (see below):

- + Argument must be fully instantiated to a term that satisfies the type.
- Argument must be unbound.
- ? Argument must be bound to a *partial term* of the indicated type. Note that a variable is a partial term for any type.
- : Argument is a meta-argument. Implies +.
- Argument is not further instantiated.
- ! Argument contains a mutable structure that may be modified using setarg/3 or nb\_setarg/3.

#### Simplest system:

- + Normally instantiated upon entry.
- Normally uninstantiated upon entry.
- ? May or may not be instantiated upon entry.

The difference between > and ? is that a ? argument may be allowed to control what the predicate does, but a > argument should not.

The type indicators can be arbitrary terms meaningful to the reader. Informal types such as atomic, integer, list and string are fine, but it is recommended that any unconventional interpretations be documented near the start of the file, program or application as appropriate. Completely informal descriptions can only work for simple and/or small projects. In more complex cases a BNF-like notation can be adopted or, better, the syntax of some formal type system, such as the ones described in (Fages and Coquery 2001; Hermenegildo 2000; Mycroft and O'Keefe 1984; Jeffery et al. 2000), can be borrowed.<sup>4</sup>

The degree of determinism of a Prolog predicate is an important part of its behaviour and so must always be documented. Doing so forces the author to consciously consider the determinism and ensures that this information is available for other users of the predicate. Furthermore, by specifying determinism in a standard-

<sup>&</sup>lt;sup>4</sup> It has not escaped our notice that the use of formal types in code documentation immediately suggests the possibility of giving the compiler access to them in order to enable useful diagnostics of all kinds if not optimized compilation. Even though we believe that adding standard ways of optionally specifying prescriptive types to Prolog would result into a significantly more powerful language, all this is completely beyond the scope of this paper.

Table 2. Predicate determinism specifiers for the template of Figure 4

det semidet	Must succeed exactly once and leave no choice-points.  Must either fail or succeed exactly once and leave no choice-points.
multi	Must succeed at least once but may leave choice-points on the last success.
nondet	May either fail or succeed any number of times and may leave choice-points on last success.

ised representation this information is then also available to automated test tools. Informally, det is used for deterministic transformations (e.g., arithmetic), semidet for tests, nondet and multi for generators.

After the predicate specification comes a clear explanation, in natural language, of what the predicate does, including special cases (in the remove\_duplicates/2 example, lists with uninstantiated elements). Normally, you should write the comment before constructing the predicate definition; if you cannot describe a predicate coherently in natural language, then you are not ready to write it.

The basic template can easily be extended to deal with multiple modes, e.g.,

```
%% age(+Name:atom, -Age:integer) is semidet
%% age(-Name:atom, +Age:integer) is nondet
%
% ...
```

Similarly, it is often convenient to document tightly related predicates with multiple arities using just one documentation block, as in

```
%% rdf_load(+File)
%% rdf_load(+File, +Options)
%
% ...
rdf_load(File) :-
   rdf_load(File, []).
rdf_load(File, Options) :-
   ...
```

When a documentation system supported by adequate tools is available, it should be given appropriate consideration. For instance, if you use SWI-Prolog, it may be advantageous to use PlDoc, the SWI-Prolog source-code documentation infrastructure (http://www.swi-prolog.org/pldoc/package/pldoc.html). The template used by PlDoc is not very different from the one introduced above. It uses the second set of argument mode specifiers, and it also provides tags such as @param,

Othrows, and Oerror for further description of a predicate's requirements and behavior.

### 4.2 Use comments to make main predicates visibly distinct from auxiliary predicates.

Auxiliary predicates exist only to continue the definition of another predicate, and are not called from anywhere else. For example, recursive loops and subordinate decision-making procedures are often placed in auxiliary predicates. While users of your code need not know about auxiliary predicates, predicate commenting can provide useful technical documentation for maintenance and repurposing.

Begin the first line of comments with % when introducing a predicate called from elsewhere in the program, but % when introducing an auxiliary predicate. For example, we can have

```
%% remove_duplicates(+List, -Processed_List) is det
%
% Removes the duplicates in List, giving Processed_List.
% ...
and
% remove_duplicates_loop(+List, -Processed_List) is det
%
% ...
```

Auxiliary predicates in Prolog often correspond to loops in other languages; the meaning of the Prolog predicate is the loop invariant, and really should be stated in a comment.

#### 4.3 Use descriptive argument names in the introductory comment.

The argument names Index, List and Elem in the introductory comment of the predicate nth0/3 below were chosen to convey its intended usage.

```
%% nth0(?Index, ?List, ?Elem)
%
% True if Elem is the Index'th element of List.
% Counting starts at 0.
nth0(Index, List, Elem) :-
```

Well-chosen argument names can also hint at the expected type of arguments, which is particularly important if explicit argument types are not specified. For example, it is reasonable to expect the argument List to be associated with some representation of a list and, in the absence of any other information, this would most likely be the usual Prolog lists representation. Likewise, Index would, in the absence of any other information, suggest a numeric argument type.

### 4.4 Argument names in the clauses should preferably be the same as in the introductory comment.

Where practical, use the same argument names in the clause definitions as used in the introductory comment, like this:

```
%% nth0(?Index, ?List, ?Elem)
%
% True if Elem is the Index'th element of List.
% Counting starts at 0.
nth0(Index, List, Elem) :-
```

However, there will be frequent occasions when this is not practical; in the following example, List in the comment corresponds to  $[X|_]$  in one clause and  $[_|T]$  in the other:

```
%% member(?Elem, ?List)
%
% True if Elem is a member of List.
member(X, [X|_]).
member(X, [_|T]) :-
    member(X, T).
```

#### 4.5 If code needs elaborate explanation, consider rewriting it.

Prolog lends itself to simple, logical programs: one predicate, one idea. The act of producing documentation is helpful also because it gives you a chance to spot design mistakes: if you use 'and' several times while describing a single predicate, that predicate is doing too many jobs and should be broken up.

```
You don't have to actively lie, just fail to keep comments as up to date with the code. 
— ROEDY GREEN (Green 2008)
```

#### 5 Language Idioms

These guidelines are particularly addressed to those who are relatively new to Prolog and accustomed to the practices of other programming languages.

#### 5.1 Predicates must be steadfast.

Any decent predicate must be "steadfast," i.e., must work correctly if its output variable already happens to be instantiated to the output value. That is,

```
?-foo(X), X = x.
```

and

?-foo(x).

must succeed under exactly the same conditions and have the same side effects. Failure to do so is only tolerable for auxiliary predicates.

### 5.2 Place arguments in the following order: inputs, intermediate results, and final results.

If you adopt a convention for the order of arguments, you have one less set of details to remember about each individual predicate. Intermediate results are those that represent a "count so far" or the like. They usually appear only in auxiliary predicates. For a much more elaborate argument-order convention, see (O'Keefe 1990, pp. 12–16).

Nonetheless, if you have a predicate with multiple arities, the argument order should not change. That is, arguments are always added on the right, regardless of whether they are for input or output.

#### 5.3 Try to limit the number of predicate arguments.

A predicate with too many arguments may be trying to do several conceptually separate jobs at once, or some of the arguments may need to be replaced with data structures containing pieces of information bundled together.

#### 5.4 Use cuts sparingly but precisely.

First think through how to do the computation *without* a cut; then add cuts to save work. For further guidance see (O'Keefe 1990, pp. 88–101).

#### 5.5 Never add a cut to correct an unknown problem.

If you don't know how a piece of code works, and you fix it blindly, you still will not know whether it is reliable after you have modified it to hide the problem.

#### 5.6 Work at the beginning of the list.

You can get to the first element of a 1000-element list in one step; getting to the last element requires 1000 steps. But do not let this or any other guideline stop you from doing the computation that you need to do. In some cases you can speed up a computation by building a list in the reverse of the order you first thought of.

#### 5.7 Avoid append where speed is important.

Generally, append is slow because it has to go all the way to the end of the first list before adding a pointer to the second. Use it sparingly in speed-critical code. But don't be silly: do not re-implement append with another name just to avoid using the original one.

One good tactic is to use append liberally in prototyping, and then change to other methods, such as difference pairs, in the finished implementation.

#### 5.8 Use difference pairs (difference lists) to achieve fast concatenation.

Suppose you want to combine the sequences  $\langle a, b, c \rangle$  and  $\langle d, e, f \rangle$  to get  $\langle a, b, c, d, e, f \rangle$ . One way is to append the lists [a,b,c] and [d,e,f].

But there's another way. Store the sequences as lists with uninstantiated tails, [a,b,c|X] and [d,e,f|Y], and keep copies of X and Y outside the lists (so you can get to them without stepping through the lists, which is what append wastes time doing).

Then just instantiate X to your second list, [d,e,f|Y]. Voilà: the first list is now [a,b,c,d,e,f|Y]; you have added material at the end by instantiating its tail. Later, by instantiating Y to something, you can add even more elements.

Difference pairs are discussed in (Covington et al. 1997) and (O'Keefe 1990).

#### 5.9 Recognize the memory advantages of tail recursion.

Tail recursion is recursion in which the recursive call is the last subgoal of the last clause, leaving no backtrack points. In this circumstance, execution can jump to the beginning of the predicate rather than creating another goal on the stack. See (Covington et al. 1997, Chapter 4).

The memory saved by tail recursion can be very important when a recursive loop has to go through thousands or millions of cycles, such as processing single characters of a large input file. When the recursion depth is measured in dozens or less, the memory saved by tail recursion is not important, and, crucially, not all computations can easily be made tail recursive. Accordingly, do not be afraid of body recursion (non-tail recursion) if you need it.

Ordinarily, in your first version of a program, you should use tail recursion if you can easily do so, but use body recursion if that makes it easier to get the code right. Then change body recursion to tail recursion if you find that there is a substantial advantage to doing so.

### 5.10 Avoid assert and retract unless you actually need to preserve information through backtracking.

Although it depends on your compiler, assert and retract are usually very slow. Their purpose is to store information that must survive backtracking. If you are merely passing intermediate results from one step of a computation to the next, use arguments.

If you have a dynamic predicate, write interface predicates for changing it instead of using 'bare' calls to assert and retract, so that your interface predicates can

check that the changes are logically correct, maintain mutexes for multiple threads, and so forth.

#### 5.11 For sorting, use mergesort or a built-in sorting algorithm.

Many Prolog compilers have efficient built-in sorting algorithms. Mergesort, an  $O(n \log n)$  sorting algorithm, is given in (Covington et al. 1997) and other books. It generally outperforms Quicksort, which is also  $O(n \log n)$  in the average case but  $O(n^2)$  in a surprisingly common worst case.

#### 6 Development, Debugging, Testing

### 6.1 Invest appropriate (not excessive) effort in the program; distinguish a prototype from a finished product.

Many Prolog programs are exploratory; when trying to figure out whether something can be done, it is reasonable to do each part of it in the easiest possible way, whether or not it is efficient.

### 6.2 The most efficient program is the one that does the right computation, not the one with the most tricks.

Using efficiency tricks, you can sometimes double the speed of a computation. By choosing a better basic algorithm, you can sometimes speed a computation up by a factor of a million.

Do not modify your code just for the sake of efficiency until you are *sure* it is actually doing the right computation. Of course there may be more than one "right" computation, so keep in mind that simplicity often leads to speed.

#### 6.3 When efficiency is critical, make tests.

Don't believe what people tell you (including us); do your own experiments. The built-in Prolog predicate statistics will tell you the time and memory used by a computation. Many Prolog compilers provide other predicates for measuring efficiency. If available in your Prolog environment, use the profiler to measure and analyze efficiency.

Use the debugger, especially a graphical one if available, to examine choice points (backtrack points) and see whether they are as you expect them to be. Careful control of choice points can dramatically reduce the memory footprint of a computation.

Results from efficiency tests are not usually portable between different Prolog environments. Prolog environment developers usually try to ensure performance of a particular construct does not deteriorate in later versions, but sometimes an alternative construct may become much faster. Make experiments in the specific version of the Prolog environment that you eventually intend to run your program

in. Where experiments lead you to chose one construct over another, document the reason for the choice by adding comments to the code.

### 6.4 Conduct experiments by writing separate small programs, not by mangling the main one.

You will often have to experiment to pin down the exact behavior of a poorly-documented built-in routine or to determine which of two computations is more efficient. Do this by writing separate small programs so that you will know exactly what you are experimenting with. Do not risk making erroneous changes in a larger program in which you have invested considerable effort. There is always the risk that if you make experimental changes, you will forget to undo them (or, worse, forget how to undo them).

#### 6.5 Look out for constructs that are almost always wrong.

These include:

- a cut at the end of the last clause of a predicate (exactly what alternatives is it supposed to eliminate?);<sup>5</sup>
- a repeat not followed by a cut (when will it stop repeating?);
- append with a one-element list as its first argument (use the element and I instead).

#### 6.6 Isolate non-portable code.

Suppose you are writing an SWI-Prolog program that passes some commands to Windows. You might be tempted to put calls such as

```
win_exec(X, normal),
...
```

all through it. Resist the temptation. Instead, define a predicate:

```
pass_command_to_windows(X) :-
    win_exec(X, normal).
```

Elsewhere in your program, call pass\_command\_to\_windows, not win\_exec. That way, when you move to another Prolog system in which win\_exec has a different name or works differently, you will only need to change one line in your program.

<sup>&</sup>lt;sup>5</sup> Such a cut can be legitimate, e.g., when a subgoal in the last clause leaves choice points that need to be discarded. Nonetheless, a predicate ending with a cut is so often a programming error that it bears careful scrutiny, and if intentional, should be commented as such (and *still* scrutinized).

#### 6.7 Isolate "magic numbers" and other constants.

If a number occurs more than once in your program, make it the argument of a fact. Instead of writing

```
X is 3.14159*Y
in one place and
Z is 3.14159*Q
somewhere else, define a fact:
pi(3.14159).
and write
pi(P),
X is P*Y
and so forth.
```

This example may seem silly because the value of  $\pi$  never changes, but if you encode 3.14159 in only one place, you don't have to worry about mis-typing it elsewhere.

And if a number in your program really does change —a number denoting an interest rate, or the maximum size of a file, or something—then it is very important to be able to update the program by changing it in just one place, without having to check for other occurrences of it.

It is common practice to define a predicate setting/2 to store constants that are likely to be changed — essentially, parameters that are hard-coded into the program, such as:

```
setting(require_login, true).
setting(timeout_milliseconds, 500).
setting(output_directory, '/home/users/student42/project/output').
...
```

### 6.8 Take the extra minute to prevent errors rather than having to find them later.

This is a basic rule of programming. Don't take the extra hour — just take the extra minute. In particular, think through loops. For example, given the predicate

```
count_up(10) :-
    !.
count_up(X) :-
    write(X),
    Y is X + 1,
    count_up(Y).
```

what is the first number printed when you count up from 1? The last number? What happens if you try to count up from 11, or from 0.5? A minute's careful thought at the right time can save you an hour of debugging on the computer.

#### 6.9 Test code at its boundaries (limits).

With count\_up above, the boundary is 10, so you should be especially attentive to situations where the argument is near 10 — slightly below 10, or exactly 10, or slightly above it.

### 6.10 Test that each loop starts correctly, advances correctly, and ends correctly.

In count\_up:

- What is the first number printed?
- After printing a particular number (such as 5), what gets printed next?
- What is the last number printed?

That's the essence of understanding any loop: you have to know where it starts, how it advances from one value to the next, and where it stops.

#### 6.11 Test every predicate by failing back into it.

It is not enough to try

?- count\_up(5).

and see what prints out. You must also try

?- count\_up(5), fail.

and see what happens when count\_up is forced to backtrack. This will show you why the first clause of count\_up has to contain a cut.

#### 6.12 Do not waste time testing for errors that will be caught anyway.

If you type

?- count\_up(five).

the program throws an arithmetic exception. That's probably okay; the important thing is that you should know that it will happen.

Do not burden your Prolog programs by testing the type of every argument of every predicate. Check only for errors that (1) are likely to occur, and (2) can be handled by your program in some useful way. For example, what happens if you execute this query?

```
?- count_up(What).
```

The most common "arguments of the wrong type" are uninstantiated arguments. Beware: they match anything!

The documentation for a predicate should make it clear which types of arguments are expected and which patterns of argument instantiation are supported. Where you think it is likely that incorrect argument types might be supplied, then add a

check to protect against this. However, to avoid cluttering up complex predicates or adversely impacting performance of recursive loops, consider adding a separate entry-point predicate that checks the arguments before invoking the original predicate. The original predicate will, of course, need to be renamed.

### 6.13 In any error situation, make the program either correct the problem or crash (not just fail).

In Prolog, failing is not the same as crashing. Crashing here means calling the built-in predicate raise\_exception/1 or, under the ISO Prolog standard, throw/1. If the exception is not caught by the program, the program will terminate with an error.

Failing is the same as answering a question "no." "No" can be a truthful and informative answer, and it is not the same as saying, "Your question does not make sense," or "I cannot compute the answer." An example:

In the last case the program should complain that georgia is not a number.

#### 6.14 Make error messages informative.

Whenever your program outputs an error message, that message must actually explain what is wrong, and, if possible, display the data that caused the problem. It should also indicate where the error was detected (e.g., by giving the predicate name). A good example:

```
mysort/3: 'x(y,z,w)' cannot be sorted because it is not a list
```

A bad example:

```
error: illegal data, wrong type
```

That doesn't say what data is the wrong type or where it was found.

Error reports in library files or modules should be related to the exported predicate they originated from, not to some internal predicate which the user doesn't know about. To achieve this, it may be necessary to add error handlers to some exported predicates just so that they can handle internal deeply generated exceptions and map them back to meaningful user-level descriptions.

#### 6.15 Master the Prolog debugger; it is simple, powerful, and portable.

The classic Prolog debugger allows you to step through a program. It works the same way in virtually all compilers and is described in (Clocksin and Mellish 2003; Covington et al. 1997) and other books.

If your Prolog environment supports graphical debugging then learn how to use it. The principles are the same as the standard command line debugger, but much more information can be displayed on-screen at once, and the program state is more readily visible.

#### 6.16 Use print for debugging.

The easiest way to see what a program does is to sprinkle it with **print** statements, so that you can see the values of the variables as the computation proceeds. This is a good tactic in all programming languages.

In Prolog, don't let backtracking confuse you. You might want to put a unique number in each print so that you can tell exactly which one produced any particular line of output.

The advantage of using print instead of write is that the output of print can be tailored and, in particular, appropriately abbreviated for debugging whereas write insists on showing you everything.

When the output from your program is not being displayed or where debug output would be hard to see for some other reason, remember that you can use print/2 to explicitly direct output to the terminal or to a log file.

If your Prolog compiler supports debug(Channel, Format, Arguments) then you can organise your comments into different categories or channels. Typical choices for channels include error, warn, info and init, but you can choose your own. Individual channels can be switched on and off as required to control the debugging output and so debug/3 statements can be left in your code permanently. If you are concerned that leaving them in might impact performance, then look for a compiler option to omit debug output commands entirely. However, if you do decide to use debug/3 then use the "p format rather than the "w format to ensure print style formatting instead of write style for the reason mentioned above.

#### 6.17 Use a revision control system.

Every serious programmer can benefit from a revision control system (version control system). For the individual developer a revision control system provides an audit trail of changes to a program; a way to retrieve previous versions; and a way to manage different branches of your code. For multiple developers, a revision control system makes collaborative work possible.

The lowest form of revision control system is to make a new backup copy of the *entire* project before every work session. This is, of course, wasteful of disk space. Proper revision control systems are available as local, client-server, and distributed systems. Commonly used packages include RCS, CVS, Subversion, and Git. There

are numerous free and commercial implementations. Choose one that suits your needs and use it.

As well as your Prolog programs, keep your test cases, documentation, data files and associated scripts under revision control too. For scientific work, repeatability is essential so make sure that you flag the version used in a published work. Many revision control systems have a web interface that is an ideal way to make your programs publicly available.

Be sure to comment out unused code instead of deleting it and relying on version control to bring it back if necessary. In no way document whether the new code was intended to supplement or completely replace the old code, or whether the old code worked at all, what was wrong with it, why it was replaced, etc. Comment it out with a lead /\* and trail \*/ rather than a [%] on each line. That way it might more easily be mistaken for live code and partially maintained.

— Roedy Green (Green 2008)

#### 7 Conclusion

It is widely acknowledged that the adoption of coding standards is one of the key factors in the success of a project: it makes it possible to cut the software maintenance costs and burden; it enables effective collaboration between developers; it may even prevent implosion of the project (due to poor readability, lack of useful documentation, use of conflicting conventions and interfaces ... plus of course the friction within the development team all this usually generates).

For a language like Prolog, coding standards are even more important, essentially due to the following factors:

- 1. powerful (and complex) language features, such as multiple modes, dynamic code, meta-programming facilities;
- 2. lack of prescriptive typing and other machine-checkable declarations;
- 3. substantial lack of sophisticated software development tools.

Despite its potential benefits, a coherent and reasonably complete set of coding guidelines for Prolog has, to the best of our knowledge, never been published. Moreover, no *de facto* standard seems to have emerged. This can be partly explained by the fact that, also due to the lack of a comprehensive language standard, the user community is fragmented into sub-communities that are centered around individual Prolog systems.

In this paper we have made a first step towards filling this gap. We have high-lighted those aspects of Prolog program development that deserve particular attention and would benefit from a disciplined approach. For each of those we have introduced a set of coding guidelines, illustrating the underlying rationale. Where alternative (or even conflicting) guidelines could achieve the desired effect, their relative merits have been discussed.

In addition to guidelines that are valuable for any Prolog programmer, this paper

contains advice that may be obvious to the more expert. This is intentional: as truly expert Prolog programmers are in very short supply (logic programming courses have been dropped in many universities), the availability and adoption of even simple coding guidelines can improve the productivity of many projects.

Of course, a good degree of subjectiveness is unavoidable on these matters. Nonetheless, we have explained the reasons why it is important to pay attention to certain aspects of Prolog and provided examples that may serve as a basis for further elaboration. Such an elaboration is unavoidable: due to differences in project purposes, environments and developer communities, a full-fledged coding standard can only be established on a per-project basis. We believe the present paper provides a useful starting point in this respect.

#### References

- Ausnit-Hood, C., Johnson, K. A., IV, R. G. P., and Opdahl, S. B., Eds. 1997. Ada 95 Quality and Style, Guideline for Professional Programmers. Lecture Notes in Computer Science, vol. 1344. Springer-Verlag, Berlin and Heidelberg, Germany.
- CLOCKSIN, W. F. AND MELLISH, C. S. 2003. *Programming in Prolog*, 5th ed. Springer, Berlin and Heidelberg, Germany.
- COVINGTON, M. A. 1994. Computer languages in type. Journal of Scholarly Publishing 26, 1, 34–41.
- COVINGTON, M. A. 2002. Some coding guidelines for Prolog. Available at http://www.ai.uga.edu/mc/plcoding.pdf.
- COVINGTON, M. A., NUTE, D., AND VELLINO, A. 1997. Prolog Programming in Depth. Prentice-Hall, Inc., Upper Saddle River, NJ, USA.
- Fages, F. and Coquery, E. 2001. Typing constraint logic programs. *Theory and Practice of Logic Programming* 1, 6, 751–777.
- GREEN, R. 1996–2008. Unmaintainable code. Available at http://mindprod.com/jgloss/unmain.html.
- HERMENEGILDO, M. V. 2000. A documentation generator for (C)LP systems. In Computational Logic: Proceedings of the First International Conference (CL 2000), J. W. Lloyd, V. Dahl, U. Furbach, M. Kerber, K.-K. Lau, C. Palamidessi, L. M. Pereira, Y. Sagiv, and P. J. Stuckey, Eds. Lecture Notes in Computer Science, vol. 1861. Springer-Verlag, London, UK, 1345–1361.
- Jeffery, D., Henderson, F., and Somogyi, Z. 2000. Type classes in Mercury. In *Proceedings of the 23rd Australasian Computer Science Conference (ACSC 2000)*. IEEE Computer Society, Canberra, Australia, 128–135.
- KERNIGHAN, B. W. AND PIKE, R. 1999. *The Practice of Programming*. Addison-Wesley Longman Publishing Co., Inc., Boston, MA, USA.
- LEDGARD, H. AND TAUER, J. 1987. Professional Software; Vol. 2: Programming Practice. Addison-Wesley Longman Publishing Co., Inc., Boston, MA, USA.
- Mycroft, A. and O'Keefe, R. A. 1984. A polymorphic type system for Prolog. *Artificial Intelligence 23*, 3, 295–307.
- O'KEEFE, R. A. 1990. The Craft of Prolog. MIT Press, Cambridge, MA, USA.
- SUN MICROSYSTEMS, INC. 1999. Code conventions for the Java programming language. Available at http://java.sun.com/docs/codeconv/.
- THOMPSON, E. M. 1893. *Handbook of Greek and Latin Palæography*. Kegan Paul, Trench, Trübner & Co, London, U.K.